Розв’язки задач першого туру

Волинської Інтернет-олімпіади з програмування 2009
1. Задача  «СНІЖИНКА»
Задача «Сніжинка» є класичною задачею на графічну побудову сніжинки. Розв’язком якої є рекурсивний обхід дерева, необхідно було вивести координати листочків. Задача ускладнена сортуванням точок.

Uses Crt, Graph;

const k=3; {коефіцієнт}
Var

Gd, Gm, n: Integer;

lon: real;

xx,yy:real;

Procedure Star(x,y,l,m,nap: Integer);

var x1,y1,i,n1: Integer;

perev: Shortint;

begin

  case nap of

     1:perev:=4;

     2:perev:=5;

     3:perev:=6;

     4:perev:=1;

     6:perev:=3;

     else perev:=0;

  end;

  for i:=1 to 6 do

  begin

  x1:=Round(l*cos(pi/3*i));

  y1:=round(l*sin(pi/3*i));

  line(x,y,x+x1,y+y1);

  n1:=m-1;

  if (n1>0) and (i<>perev) then star(x+x1,y+y1,round(l/k),n1,i);

  end;

end;

Begin

Write('l='); Readln(lon);

Write('n='); ReadLn(n);

Writeln('x='); Readln(xx);

Writeln('y='); Readln(yy);

 Gd:=Detect;

 InitGraph(Gd, Gm, '');

 If GraphResult <> grOk Then Halt(1);

 Star(round(xx),Round(yy),Round(lon),n,0);

 ReadLn;

 CloseGraph;

End.
Задача  «СТРАТЕГІЯ»
Міста утворюють дерево. Потрібно “підвісити” це дерево за будь-яку із вершин, і таким чином отримати дерево з виділеним коренем, при цьому у кожної вершини, крім кореня, буде відомий її батько. Потім необхідно застосувати метод динамічного  програмування: для кожної вершини V і для кожного числа T від 1 до P підрахуємо, яку  мінімальну кількість доріг повинно бути захвачено, щоб в піддереві з коренем у V компоненті зв’язності (максимальна зв’язна частина), що містить вершину V, складалась із T вершин — назвемо це число Best[V,T]. Будемо підраховувати значення Best, ідучи по вершинам від листків до кореня. Якщо для всіх гілок деякої вершини V ці значення вже підраховані, ми можемо підрахувати їх і для цієї вершини. Для цього розглянемо, як встроєна деяка компонента зв’язності із T вершин, яка містить вершину V. Вона складається із вершини V, а також із компонент, що починаються в гілках цієї вершини, і мають сумарний розмір T–1. Тоді перебираючи всі розбиття числа T–1 на невід’ємні цілі складові T–1 = a1+a2+ … +am (їх число m повинно бути рівне числу гілок вершини V), ми беремо в якості чергового кандидата суму Best[V1,a1] + Best[V2,a2] + … + Best[Vm,am] (де V1, V2, …, Vm — гілки вершини V), причому якщо якесь ai дорівнює 0, то ми рахуємо Best[Vi,ai] = 1. Далі із всіх вищезгаданих кандидатів ми вибираємо найменший і записуємо в Best[V,T]. При такому розв’язку досить довгий час може займати перебір всіх розклад числа T – 1 на декілька складових, але цього частково можна уникнути. Після того, як значення Best підраховані, залишається отримати із них відповідь задачі. А саме, для того щоб відрізати від останніх міст групу із P вершин з коренем у вершину V може залишити Best[V,P] доріг, плюс іще одну дорогу у випадку, якщо вершина V не є коренем дерева (а саме, дорогу, що з’єднує піддерево з коренем у V з деревом, що залишилось). Природно, тепер необхідно вибрати із всіх таких способів той, який  потребує захвату мінімальної кількості доріг.

Const

  MaxN=152;

  Infinity=200;

Var

  I,N,P,M:LongInt;

  Country:Array[1..MaxN]Of Record A,B:LongInt; End;

  Min:Array[1..MaxN,1..MaxN]Of Byte;

  Visited:Array[1..MaxN]Of Boolean;

  Res:LongInt;

Procedure DFS(Start:LongInt);

Var

  I,J,K:LongInt;

  Next:LongInt;

Begin

  Visited[Start]:=True;

  FillChar(Min[Start],SizeOf(Min[Start]),Infinity);

  Min[Start][1]:=0;

  For I:=1 To M Do

    If (Country[I].A=Start) Or (Country[I].B=Start) Then Begin

      If Country[I].A=Start Then

        Next:=Country[I].B

      Else

        Next:=Country[I].A;

      If Not Visited[Next] Then Begin

        DFS(Next);

        For J:=N DownTo 1 Do If Min[Start][J]<Infinity Then Begin

          For K:=1 To N Do

            If Min[Next][K]<Infinity Then

              If Min[Start][J+K]>Min[Start][J]+Min[Next][K] Then

                Min[Start][J+K]:=Min[Start][J]+Min[Next][K];

          Inc(Min[Start][J]);

        End;

      End;

    End;

End;

Begin

  Assign(Input,'town.dat');

  ReSet(Input);

  Read(N,P);

  For M:=1 to N-1 do

    Read(Country[M].A,Country[M].B);

  M:=N-1;

  Close(Input);

  FillChar(Visited,SizeOf(Visited),False);

  DFS(1);

  Res:=Min[1][P];

  For I:=2 To N Do

    If Min[I][P]+1<Res Then

      Res:=Min[I][P]+1;

  Assign(Output,'town.sol');

  ReWrite(Output);

  WriteLn(Res);

  Close(Output);

End.

